2 Divisors, Polytopes, and Posititivity

How are divisors and polytopes related? The dictionary between polyhedral combinatorics and algebraic geometry contains more.

A *Weil divisor* is a formal sum with integer coefficients of reduced and irreducible closed subschemes of codimension one. A *Cartier divisor* is locally principal (locally defined by a single rational function).

Divisors. Throughout, assume that the normal toric variety X has no torus factors and set $n := |\Sigma(1)|$. There is a commutative diagram of abelian groups with exact rows

$$\begin{tikzcd}
0 \arrow[r] & M \arrow[r] \arrow[d] & \mathrm{CDiv}_T(X) \arrow[r] \arrow[d] & \mathrm{Pic}(X) \arrow[r] \arrow[d] & 0 \\
0 \arrow[r] & M \arrow[r] & \mathbb{Z}^n \cong \bigoplus_{j=0}^{n-1} \mathbb{Z} D_j := \mathrm{Div}_T(X) \arrow[r] & \mathrm{Cl}(X) \arrow[r] & 0\,.
\end{tikzcd}$$

The bottom row only depends on the rays in the fan.

Proposition. *For all X, the following are equivalent:*
(a) *Every Weil divisor is Cartier:* $\mathrm{Div}_T(X) = \mathrm{CDiv}_T(X)$*;*
(b) $\mathrm{Cl}(X) = \mathrm{Pic}(X)$*;*
(c) *X is smooth.*

Proposition. *For all X, the following are equivalent:*
(a) *Every Weil divisor has a positive integer multiple that is Cartier.*
(b) $\mathrm{Pic}(X)$ *has finite index in* $\mathrm{Cl}(X)$.
(c) *X is simplicial.*

Theorem. *For any Cartier divisor D on X, the following are equivalent:*
(a) *D is basepoint free,*
(b) *the line bundle $\mathscr{O}_X(D)$ is globally generated,*
(c) *D is nef ($D \cdot C \geqslant 0$ for every irreducible complete curve $C \subseteq X$),*
(d) *$D \cdot C \geqslant 0$ for all T_N-invariant irreducible complete curves $C \subseteq X$.*

Every Cartier divisor determines to invertible sheaf $\mathscr{O}_X(D)$ and every line bundle arises from a Cartier divisor.

Primitive Relations. A subset of $\gamma \subseteq \{0, 1, \dots, n{-}1\}$ is called a *primitive collection* is the rays indexed by γ generate a minimal noncone.

How do we interpret the maximal cones in a fan geometrically?

Proposition. *The irreducible components of the irrelevant set* $\mathrm{V}(B)$ *are defined by the prime ideals generated by primitive collections. In other words, the monomial ideal B is the Alexander dual of the Stanley–Reisner ideal of Σ.*

Definition. Let $\gamma \subseteq \{0, 1, \dots, n-1\}$ be a primitive collection in a complete simplicial fan Σ. The vector $\mathbf{v}_{\gamma_0} + \mathbf{v}_{\gamma_1} + \cdots + \mathbf{v}_{\gamma_k}$ lies in the relative interior of a unique cone $\sigma \in \Sigma$, so

$$\mathbf{v}_{\gamma_0} + \mathbf{v}_{\gamma_1} + \cdots + \mathbf{v}_{\gamma_k} = \sum_{j \in \sigma(1)} c_j \, \mathbf{v}_j \quad \text{where } c_j \in \mathbb{Q} \text{ and } c_j \geqslant 0.$$

The coefficient vector of the relation $\mathbf{v}_{\gamma_0} + \mathbf{v}_{\gamma_1} + \cdots + \mathbf{v}_{\gamma_k} - \sum_{j \in \sigma(1)} c_j \, \mathbf{v}_j$ is $r(\gamma) \in \mathbb{R}^n$.

Theorem. *When X is a projective simplicial toric variety, the Mori cone of X is* $\mathrm{cone}\big(r(\gamma) \mid \gamma \text{ is a primitive collection}\big)$. *The nef cone* $\mathrm{Nef}(X)$ *is dual to the Mori cone, so the primitive relations correspond to supporting hyperplanes.*

The interior of the nef cone can be identified with the Kähler cone.

Copyright © 2026 by Gregory G. Smith

Polytopes. Rather than fans in $N_{\mathbb{R}}$, considers polytopes in $M_{\mathbb{R}}$.

Theorem. *There is a bijection between the set of full-dimensional lattice polytopes in $M_{\mathbb{R}}$ and the set of pairs (X, D) where X is a projective toric variety and D is an ample divisor on X.*

Proof Idea. The (innerward) normal fan of the polytope defines X, and the polytope $P = \{\mathbf{u} \in M_{\mathbb{R}} \mid \langle \mathbf{u}, \mathbf{v}_j \rangle \geqslant -a_j \text{ for all } 0 \leqslant j \leqslant n-1\}$ corresponds to $D_P = a_0 D_0 + a_1 D_1 + \cdots + a_{n-1} D_{n-1}$. □

Having fixed a projective toric variety X, every Weil divisor determines to a (possibly empty) polytope.

Theorem. *A normal toric variety is projective if and only if its fan is the normal fan of a polytope.*

More generally, we have

Proposition. *Let $S := \mathbb{C}[x_0, x_1, \dots, x_{n-1}]$ be the Cox ring of X. For all $\alpha \in \mathrm{Cl}(X)$, we have $S_\alpha \cong H^0(X, \mathscr{O}_X(\alpha))$; the monomials in $\mathbb{C}$-vector space S_α correspond to the T_N-invariant sections of reflexive sheaf $\mathscr{O}_X(\alpha)$ and the lattice points in the polytope.*

Definition. A complete normal toric variety X is *Fano* if its anticanonical divisor $-K_X = D_0 + D_1 + \cdots + D_{n-1}$ is Cartier and ample.

Theorem. *A normal toric variety X is* Fano *if and only if the polytope associated to its anticanonical divisor $-K_X$ is reflexive.*

A lattice polytope is *reflexive* if its polar dual is also a lattice polytope.

Computational Examples in *Macaulay2*.

```
Macaulay2, version 1.26.06
Type "help" to see useful commands
i1 : needsPackage "NormalToricVarieties";
i2 : X0 = normalToricVariety(
          {{1,1,1},{-1,1,1},{1,-1,1},{-1,-1,1},{1,1,-1},{-1,1,-1},
              {1,-1,-1},{-1,-1,-1}},
          {{0,1,2,3},{0,1,4,5},{0,2,4,6},{1,3,5,7},{2,3,6,7},{4,5,6,7}});
i3 : isWellDefined X0 and isProjective X0 and not isSmooth X0
o3 = true
i4 : cartierDivisorGroup X0
       4
o4 = ZZ
o4 : ZZ-module, free
i5 : picardGroup X0
       1
o5 = ZZ
o5 : ZZ-module, free
i6 : classGroup X0
o6 = cokernel | 2 0 |
              | 0 2 |
              | 0 0 |
              | 0 0 |
              | 0 0 |
              | 0 0 |
              | 0 0 |
                               7
o6 : ZZ-module, quotient of ZZ
```

Copyright © 2026 by Gregory G. Smith

```
i7 : weilDivisorGroup X0
         8
o7 = ZZ
o7 : ZZ-module, free
i8 : fromPicToCl X0
o8 = | 0 |
     | 0 |
     | 0 |
     | 2 |
     | 2 |
     | 2 |
     | 2 |
o8 : Matrix
i9 : fromWDivToCl X0
o9 = | 1  0  1  0 0 0 0 0 |
     | 1  1  0  0 0 0 0 0 |
     | 1  -1 -1 1 0 0 0 0 |
     | -1 1  1  0 1 0 0 0 |
     | 0  0  1  0 0 1 0 0 |
     | 0  1  0  0 0 0 1 0 |
     | 1  0  0  0 0 0 0 1 |
o9 : Matrix
i10 : fromCDivToWDiv X0
o10 = | 1  1  1  1 |
      | -1 1  1  1 |
      | 1  -1 1  1 |
      | -1 -1 1  1 |
      | 1  1  -1 1 |
      | -1 1  -1 1 |
      | 1  -1 -1 1 |
      | -1 -1 -1 1 |
                 8      4
o10 : Matrix ZZ  <-- ZZ
i11 : fromWDivToCl X0 * fromCDivToWDiv X0 == fromPicToCl X0 * fromCDivToPic X0
o11 = true
i12 : Y = makeSmooth X0
o12 = Y
o12 : NormalToricVariety
i13 : # rays Y, # rays X0
o13 = (26, 8)
o13 : Sequence
i14 : # max Y, # max X0
o14 = (48, 6)
o14 : Sequence
i15 : isWellDefined Y and isSmooth Y
o15 = true
i17 :  D = smallAmpleToricDivisor(2,13)
o17 = D  + 2*D  + 3*D  + 3*D
       0      1      4      5
o17 : ToricDivisor on normalToricVariety ({{1,0},{-1,0},{0,1},{1,1},{0,-1},...
i19 : X1 = variety D
o19 = X1
o19 : NormalToricVariety
i20 : isAmple D
o20 = true
```

Copyright © 2026 by Gregory G. Smith

```
i21 : isAmple(X1_0+X1_3)
o21 = false
i22 : isNef(X1_0+X1_3)
o22 = true
i23 : K = toricDivisor X1
o23 = - X1  - X1  - X1  - X1  - X1  - X1
          0     1     2     3     4     5
o23 : ToricDivisor on X1
i24 : isAmple(-K)
o24 = true
i25 : S = ring X1;
i26 : transpose matrix degrees S
o26 = | 1  1 0  0 0 0 |
      | -1 0 -1 1 0 0 |
      | 0  0 1  0 1 0 |
      | 1  0 1  0 0 1 |
                4       6
o26 : Matrix ZZ  <-- ZZ
i27 : nefGenerators X1
o27 = | 1 0 1 1  1  |
      | 0 0 0 -1 -1 |
      | 0 1 1 1  1  |
      | 1 1 1 1  2  |
                4       5
o27 : Matrix ZZ  <-- ZZ
i28 : LP = latticePoints polytope D
o28 = {| -1 |,| -1 |,| -1 |,0,| 0 |,| 0 |,| 0 |,| 1 |,| 1 |,| 1 |,| 2 |,| 2 |}
       | 1  | | 2  | | 3  |   | 1 | | 2 | | 3 | | 0 | | 1 | | 2 | | 0 | | 1 |
o28 : List
i29 : #LP
o29 = 12
i30 : basis(degree D, S)
o30 = | x_0^3x_2x_3^3x_4^2 x_0^3x_3^2x_4^3x_5 x_0^2x_1x_2^2x_3^3x_4 ...
               1       12
o30 : Matrix S  <-- S
i31 : basis(degree (-K), S)
o31 = | x_0^2x_2x_3^2x_4 x_0^2x_3x_4^2x_5 x_0x_1x_2^2x_3^2 ...
               1       7
o31 : Matrix S  <-- S
i32 : latticePoints polytope (-K)
o32 = {| -1 |, | -1 |, | 0  |, 0, | 0 |, | 1  |, | 1 |}
       | 0  |  | 1  |  | -1 |     | 1 |  | -1 |  | 0 |
o32 : List
i33 : ideal X1
o33 = ideal (x x x x , x x x x , x x x x , x x x x , x x x x , x x x x )
              1 2 4 5   1 2 3 5   0 3 4 5   0 2 3 4   0 1 4 5   0 1 2 3
o33 : Ideal of S
i34 : dual monomialIdeal X1
o34 = monomialIdeal (x x , x x , x x , x x , x x , x x , x x , x x , x x )
                      0 1   0 2   1 3   1 4   2 4   3 4   0 5   2 5   3 5
o34 : MonomialIdeal of S
i35 : primaryDecomposition ideal X1
o35 = {ideal(x ,x ),ideal(x ,x ),ideal(x ,x ),ideal(x ,x ),ideal(x ,x ),...
              1  0         2  0         5  0         3  1         4  1
o35 : List
i36 : P = polytope D;
```

Copyright © 2026 by Gregory G. Smith

```
i37 : X' = normalToricVariety P;
i38 : set rays X1 == set rays X'
o38 = true
i39 : X2 = normalToricVariety(
          {{1,0,0},{0,1,0},{0,0,1},{0,-1,2},{0,0,-1},{-1,1,-1},
              {-1,0,-1},{-1,-1,0}},
          {{0,1,2},{0,2,3},{0,3,4},{0,4,5},{0,1,5},{1,2,7},{2,3,7},
              {3,4,7},{4,5,6},{4,6,7},{5,6,7},{1,5,7}});
i40 : isWellDefined X2 and not isProjective X2 and isComplete X2 and isSmooth X2
o40 = true
i41 : X3 = smoothFanoToricVariety(4,123);
i43 : isSmooth X3 and isProjective X3
o43 = true
i44 : isAmple( - toricDivisor X3)
o44 = true
i45 : #latticePoints polytope (- toricDivisor X3)
o45 = 55
i46 : nefGenerators X3
o46 = | 1 0 0 0 1 0 |
      | 0 1 1 0 0 0 |
      | 0 0 1 0 0 1 |
      | 0 0 0 1 0 1 |
      | 0 0 0 0 1 1 |
                 5        6
o46 : Matrix ZZ  <-- ZZ
```

Copyright © 2026 by Gregory G. Smith